

```

1 *****
2 *
3 *      WPMerlinTab20 - An AppleWorks Init
4 *
5 *      - When loading Merlin Source Code Files
6 *      possessing both a ProDOS File Type
7 *      of 'TXT'/$04 AND ending in '.S'
8 *      into the Word Processor, this Init
9 *      edits the default <TAB> ruler for
10 *      the file and sets the <TAB> stops
11 *      at columns #15, #22 and #39 to
12 *      present a display similar to that
13 *      of the Merlin Editor.
14 *
15 *
16 *      Version 2.0 (for AppleWorks Version 5.1)
17 *      (c) 2017 Hugh Hood
18 *
19 *      - Seg $28 (Load Files) is patch-in segment.
20 *
21 *****
22
23          TR          ADR          ; truncate bank address
24
25          XC          ; enable 65C02 code
26
27 * Equates *
28
29 AWWVersion      EQU          $1003          ; $33/51 = 5.1 / $28/40 = 4.0 /
30                                     ; $1E/30 = 3.0
31 MvLeftRtn      EQU          $1148          ; memory move / follow with TO/FROM/LENGTH
32 StrMvRtn       EQU          $116C          ; string move / follow with TO/FROM
33 imSavePatch    EQU          $3006          ; Patch Manager save routine in SEG.IM
34 InitAdr        EQU          $4000          ; load address for Init files
35 OpenParamName  EQU          $51BC          ; Filename Used by OpenInput Routine
36 PatchPoint1    EQU          $6432          ; first patch point in SEG $28
37 WPClearHeader  EQU          $6E05          ; prepare for new <TAB> ruler
38 TabStLen       EQU          $7C10          ; $4F/79 - length of tab ruler
39 TabStops       EQU          $7C11          ; standard tab ruler stops/characters
40 MMInDoc        EQU          $7C68          ; AWP file flag byte:
41                                     ; bit 0 (%00000001)/$01 = Mail Merge (AWP)
42                                     ; bit 2/5 (%00100100)/$24 = Merlin Source
43                                     ; bit 3/5 (%00101000)/$28 = Unix TEXT
44                                     ; bit 4/5 (%00110000)/$30 = MS-DOS TEXT
45                                     ; bit 5 (%00100000)/$20 = TEXT
46 Patch2Run      EQU          $6D00          ; final destination for new code
47                                     ; NOTE: Seg $28 runs from $5200 - $65F6,
48                                     ; and uses some space following
49                                     ; $6600.
50 *****
51 * IMPORTANT NOTE: File Tags are normally loaded (one-by-one) from the file into the
52 *      unused space from $6600-$6DFF prior to being saved to Desktop
53 *      Memory with a pointer stored in the Tag Table. This space will
54 *      support a maximum Tag size of $800/2048 bytes. This init will
55 *      reduce the maximum possible Tag size to $700/1792 bytes.
56 *      Fortunately, all (3) of the officially-registered AppleWorks file

```



```

112
113 HookBytes      HEX          0000          ; first bytes for $28 Patch
114
115                LDA          #$4C          ; JMP instruction
116                STA          PatchPoint1   ; $6432 in SEG $28
117                LDA          #Patch2Run    ; low byte of new code
118                STA          PatchPoint1+1
119                LDA          #>Patch2Run   ; high byte of new code
120                STA          PatchPoint1+2
121
122
123 MoveCall        JSR          MvLeftRtn     ; move new code to run location
124                DA          Patch2Run      ; ($6D00)
125                DA          MoveStart      ;
126                DA          Patch2End-Patch2Run
127
128                RTS                    ; patch hook-in done
129
130 Patch1End       EQU          *
131
132
133 **-----
134
135 MoveStart       EQU          *          ; (will be $BBxx)
136
137                ORG          Patch2Run     ; ($6D00)
138
139
140 * Check Text file for '.S' Extension - Merlin <CR>
141
142 PatchStart      JSR          WPClearHeader ; clear header (for new <TAB> ruler)
143 SetTXT          LDA          #%00100000    ; #$20
144                STA          MMInDoc
145
146                LDX          OpenParamName ; length of full pathname
147                LDA          OpenParamName,X
148                CMP          #'S'         ; is last character 'S'?
149                BNE          GoBack        ; not Merlin
150                LDA          OpenParamName-1,X
151                CMP          #'.'         ; is second-to-last '.'?
152                BNE          GoBack        ; not Merlin
153
154                JSR          StrMvRtn      ;
155                DA          TabStLen      ; Ruler starts at $7C10
156                DA          MerlinRuler   ; special Merlin ruler
157
158 SetMLN          LDA          #%00100100    ; #$24 (keep TXT/#$20 Bit)
159                STA          MMInDoc
160
161 GoBack          JMP          PatchPoint1+8 ; go back to original code ($643A)
162
163 **-----
164
165
166 * Modify this <TAB> Ruler to Suit your needs for Merlin Source Display
167

```

